
ABSTRACT

This paper presents an algorithm that simulates the calls of
the Hyalessa maculaticollis cicada for musical use. Writ-
ten in SuperCollider, its input parameters enable real-time
control of the insect call phase, loudness, and perceived
musical pitch. To this end, the anatomical mechanics of
the tymbal muscles, tymbal apodeme, tymbal ribs, tymbal
plate, abdominal air sac, tympana, and opercula are phys-
ically modeled. This also includes decoherence, follow-
ing the hypothesis that it, in H. maculaticollis, might ex-
plain the change in timbre apparent during the final phase
of a call sequence. Overall, the algorithm seems to il-
lustrate three main points regarding the trade-offs encoun-
tered when modeling bioacoustics for tonal use: that it may
be necessary to prioritize musical requirements over real-
istic physical modeling at many stages of design and im-
plementation; that the resulting adjustments may revolve
around having physical modeling perceptually yield sonic
events that are well-pitched, single-attack, single-source,
and timbrally expressive; that the pitch-adjusted simulation
of resonating bodies may fail musically precisely when it
succeeds physically, by inducing the perception of differ-
ent sound sources for different pitches. Audio examples
are included, and the source code is structured and docu-
mented so as to support the further development of cicada
bioacoustics for musical use.

1. INTRODUCTION

Reflecting on immersive sound, music and computing – the
theme of this year’s conference – one way in which forms
of instrumental control can be designed so as to recogniz-
ably tie musical pieces in which they are used to specific
geographical regions, time periods, or circumstances, is to
have the heard sounds that they produce mimick those that
occur in the corresponding, natural environments.

Simultaneously, however, another musical requirement
may well be that these heard sounds also should have a
clearly perceived pitch which can be changed in real time
– e.g. to enable melodies; and that across such pitch or
other control changes, the resulting heard sound will still,
via the process of human Auditory Scene Analysis (ASA,

Figure 1. An H. maculaticollis individual in its natural en-
vironment, showing its tymbal, abdomen & operculum. 2

see Bregman in [1]), be perceived as coming from a single
source – e.g. to help enable use in counterpoint.

This paper presents an algorithm that attempts to balance
these three musical requirements for the case of H. mac-
ulaticollis, a species of cicada that is native to the Korean
Peninsula.

Admittedly, when looking for a suitable starting point
to investigate physical modeling versus musical require-
ments, choosing tonal use may not immediately seem like
an obvious choice. After all, are there not more recently
introduced musical paradigms available, subject to more
dynamic ongoing developments?

However, if we do not want to first have to debate the con-
tents of a specific musical paradigm, and instead want to
focus on its interplay with physical modeling, this seeming
disadvantage becomes an advantage: The requirements for
tonal use are relatively well-understood, static, and non-
controversial. For example, it seems relatively clear-cut
that tonal aesthetics will usually be violated if sound syn-
thesis does not provide musical voices with accurately and
precisely controlled perceived pitches.

2 Cropped and annotated version of a photograph by 더 그레이스,
licensed under https://creativecommons.org/licenses/
by-nc/4.0/ and obtained via [2].

1.1 Approach: Physical Modeling of Anatomical
Mechanics, Informed by the Analysis of Field
Recordings

Earlier work on synthesizing cicada sound includes [3] by
Georgaki and Queiroz. Here, a different species, Cicada
orni, is studied, which produces calls that in their timbre
and rhythm are quite different from those made by H. mac-
ulaticollis. Still, this work clearly illustrates how, generally
speaking, cicada sound synthesis can be based on the anal-
ysis of field audio recordings. For the present work, two
field recordings in particular, kindly made publicly avail-
able at [4] and [5], have been used as a reference through-
out development.

However, a choice was made to also base development on
biological knowledge of anatomical mechanics, motivated
by the potential of H. maculaticollis to provide a detailed
case study of where along the causal structure of bioacous-
tics trade-offs between realistic modeling and musical ap-
plicability may lie.

For gaining the required biological knowledge, Pollack
provided a very useful starting point in [6], giving a clear
introductory overview of the anatomical components and
mechanics of cicada sound production, based on earlier
work by others, while placing this within a wider context
of insect bioacoustics.

H. maculaticollis can then be approached as a specific in-
stance of cicadas and their sound production, with some
of the key anatomical components illustrated in Figure 1.
Here, an individual of the species is shown residing in its
natural environment, with the anatomical locations of its
left tymbal, abdomen, and left operculum indicated. The
tymbal is an organ containing ribs that can be sequentially
pulled inward by the tymbal muscles, via the internal con-
necting tissue of the tymbal apodeme. The resulting re-
verberations go into other parts of the tymbal; into the air
sac that is contained within the abdomen; and into the sur-
roundings, especially via the left and right tympana. Here,
each tympanum, while itself an acoustically open mem-
brane, is variably covered by its corresponding operculum.

In order to start, based on the above general context, on
the development of a concrete algorithm, the information
provided by Bennet-Clark and Young in [7] and [8] has
been crucial: Detailed description and measurements re-
garding the relevant anatomy and its mechanics and acous-
tics are provided, as well as a rigorously tested model of
the resulting sound generation, including abdominal res-
onation. Although this is based on the study of other
species of cicada, especially Cyclochila australasiae, re-
sults are evaluated as a model for sound production in ci-
cadas in general, and seemed a suitable starting point when
attempting to develop the algorithmic simulation of cicada
calls by H. maculaticollis.

Further on, the work by Smyth and Smith reported in [9]
and [10] was helpful: As a part of reported work on a
novel musical instrument, two general digital filter types
are mentioned that can be used when implementing the
modeling of C. australasiae tymbal and abdominal res-
onation, respectively.

2. DESIGN AND IMPLEMENTATION OF THE
ALGORITHM

The algorithm was written from scratch in SuperCollider,
and is included in annotated source format in the Appen-
dices. It has six input arguments, each of which can be
changed in real time: a_gate_bit, loudness_nrm,
t_mae_trig, t_mae_retrig, t_mi_trig, and
freq_hz.

The function of audio-rate argument a_gate_bit is to
enable graceful termination of the algorithm. This both in
terms of the amplitudes of its stereo audio output, and in
terms of the release of claimed computational resources.
Via the control-rate argument loudness_nrm, the per-
ceived loudness of algorithm output can be set, along a
scale from 0.0 to 1.0.

Unlike C. australasiae, perceptually, the calls of H. mac-
ulaticollis seem to follow a three-part structure, consisting
of an initial phase, a repeated middle phase, and a final
phase. The arguments t_mae_trig, t_mae_retrig,
and t_mi_trig make each of these phases, respectively,
subject to real-time control: A 1.0 impulse to any of these
trigger-type arguments signals the algorithm to transition
to the corresponding phase of the call. Here, parts of
the transliterated onomatopoeic Korean word for cicada,
“maemi”, are used as a mnemonic, to indicate which phase
of the insect call each argument refers to. For an example
of the use of these input arguments to control a complete
call sequence, see the waveform at the bottom of Figure 3.

Finally, the remaining, control-rate argument freq_hz
is used to relay, as a frequency in hertz, the intended per-
ceived musical pitch that algorithm output should induce in
the listener. Here, we encounter the first trade-off between
realistic physical modeling and musical requirements: Ci-
cada song frequencies are emphatically present at thou-
sands of hertz, but the modeled input range for freq_hz
was set to the octave from E4 to E5. This so that the heard
pitches would still lie within one of the main ranges of
tonal music: that of the soprano, the highest human vocal
range.

The subsections below will give an introduction to the
algorithm itself. For reasons of clarity, this will be done
by traversing various parts of the insect’s sound-generating
anatomy in a causal order. The information on cicada
anatomy and sound production that is used will be based
on [7] and [8], unless explicitly stated otherwise.

2.1 The Tymbal Muscles: Contraction & Relaxation

During cicada song, the tymbal muscles, connected to
the left and right tymbal, go through repeated cycles of
contracting inward and then relaxing. In the algorithm,
this was modeled in the tymbalMuscle_trig signal, a
unipolar waveform consisting of single-sample impulses,
one for each contraction cycle. During the segments of
cicada song where the tymbal ribs buckle in tight syn-
chronicity (and thereby produce a tightly coherent pulse
train), each tymbalMuscle_trig impulse models the
moment in time when contraction causes the first tymbal
rib to start to buckle.



Moving on to computing the frequency of muscle con-
tractions during such segments, the available information
on C. australasiae presented a clear problem: Live con-
trol of the algorithm should, as discussed above, yield per-
ceived musical pitches lying between 329 and 660 Hz.
This, however, is both above the 100-200 Hz range of nat-
ural tymbal muscle contractions, and below the resulting
natural song frequency of around 4300 Hz.

To resolve this, a musically-motivated choice was made
to have each contraction always cause exactly 4 buckling
events (see Section 2.3 for details on their nature). The
idea was that, in this way, muscle contraction modeled at a
mostly natural rate could cause strong overtones at the pre-
cise pitch frequency specified, with the insect song over-
tones then present above that (while not being perceptually
dominant as the pitch). Accordingly, the starting point for
the computation of the tymbalMuscleFreq_hz signal
was a division by 4 of freq_hz.

Beyond this, computing tymbalMuscleFreq_hz
was based on analysis of the H. maculaticollis field record-
ings. Here, a first observation was that during the ini-
tial and middle phases of ordinary call sequences, tymbal
muscle frequency seemed to be at mostly higher levels as-
sociated with tightly synchronous tymbal rib buckling, to
then end at lower levels during the final phase. For the al-
gorithm, a musical choice was then made to model these
lower levels so as to be a whole number of octaves lower.
The idea was that this might help harmonically: First, to
clearly connect the final call phase to the initially heard
pitch; and second, to avoid unintended dissonances during
use with other musical voices in a piece. A subsequent
choice, returning to primarily considering physical model-
ing, was then to go a single whole octave lower, but not
more, as the latter seemed to yield results sounding too
similar to a general end of cicada calling.

After this, a more involved physical modeling choice
was to simulate the control of tymbal muscle frequency
in further detail by implementing a system of envelopes.
Here, each call phase was modeled by a separate enve-
lope, with each envelope starting or terminating based on
the live input from t_mae_trig, t_mae_retrig, and
t_mi_trig; and with all envelope output combining into
a single, continuous signal.

Within the individual envelopes, the target levels, dura-
tions, and curvatures defining each envelope segment were
– except when based on the octave difference mentioned
above – directly derived from measurements of features
present in the field recording obtained from [4]. Where
possible, this was done by comparing a set of 12 adjacent
harmonics that were evident in spectrograms sampled at
11025 Hz. These were visible in the 3.3-5.3 kHz range
of the field recording, and in the 0.9-2.7 kHz range of al-
gorithm output (when testing at pitch E5). The results of
these measurements have been explicitly labeled and un-
ambiguously structured in the algorithm, using a series of
15 scalar constants that define the envelopes, and 4 signal
variables that contain their output.

While doing the above, a musical choice was made to not
mimick a general downward trend seen spanning across

the harmonics of repeated middle call phases in the field
recording. This to help avoid algorithm output going out
of tune during tonal use.

A final, speculative addition to tymbal muscle frequency
control was then to give an explicit presence to the idea
of it also being subject to further, possibly unknown but
smaller additive effects. This was done by raising or lower-
ing the computed frequency by a semi-normally distributed
random value, updated at least once every tymbal muscle
contraction. The peak amplitude for this was limited to 10
cents, however, as going higher resulted in ripples in mus-
cle frequency harmonics that were not visible in the field
recordings. Perceptually, the result then seemed to yield at
most a slight variation. In any case, this extra computation
can be effectively removed from the algorithm by changing
a single line of code.

Example output of the live computation of
tymbalMuscleFreq_hz during a full call sequence at
pitch E5 can be seen in the waveform placed second-lowest
in Figure 3. The corresponding tymbalMuscle_trig
signal can be heard in of the Appen-
dices.

2.2 The Tymbal Apodeme: Pull & Release

Each muscle contraction cycle, the tymbal apodeme, be-
ing the connecting tissue between a tymbal muscle and the
tymbal itself, is pulled inward with variable force, and then
released. Relevant results from the literature on C. aus-
tralasiae here include that for tymbal rib buckling events,
their phase coherence as a pulse train is supported by the
mechanics of the insect’s anatomy from tymbal apodeme
to abdominal air sac; and that at higher apodeme pulling
speeds, it appears “that the tymbal buckling mechanism
offers good coupling between the vibrations of the ribs so
that the rapidly excited tymbal inevitably produces a co-
herent waveform”.

This information influenced physical modeling in
the algorithm via two main points of interpretation:
First, that we will assume that apodeme pulling
speed is related to tymbal muscle contraction fre-
quency. As a first approximation for modeling this,
in the algorithm, the first stage of computing the
tymbalApodeme_pullingSpeed_nrm signal has
been to make it directly proportional to the output of the
tymbal muscle frequency envelopes.

The second point of interpretation was then that, above
a certain threshold of apodeme pulling speed, tym-
bal mechanics will pull buckling into (the same) tight
coherence. In the algorithm, this was implemented
by modeling tymbalApodeme_pullingSpeed_nrm
as a normalized value, and then clipping and scal-
ing it to a maximum corresponding to the threshold.
The concrete value for the threshold was then cho-
sen so that tymbalApodeme_pullingSpeed_nrm
would mostly be at its maximum level during the ini-
tial and middle phases of H. maculaticollis call se-
quences – the parts where in the field recordings, tym-
bal muscle contraction frequency mostly seemed to be
at higher levels associated with tightly synchronous




tymbal rib buckling. The resulting computation of
tymbalApodeme_pullingSpeed_nrm is illustrated
by the waveform placed third-lowest in Figure 3.

2.3 The Tymbal Ribs: In-Buckling & Out-Buckling

2.3.1 Modeling the Individual Tymbal Rib Buckling
Events

The literature indicates that for C. australasiae, during nor-
mal song, a single pull-and-release cycle by the tymbal
muscle and apodeme results in the first 2, sometimes 3
tymbal ribs sequentially buckling inward, from posterior
to anterior (while the 4th rib only buckles during “protest
song”); and that after this, there is a simultaneous buckling
back outward.

For the algorithm, as a trade-off between physical model-
ing and musical requirements, the choice was made to al-
ways model 4 buckling events per muscle contraction (this
has been motivated in Section 2.1): the in-buckling of tym-
bal ribs 1-3, followed by their combined out-buckling. The
associated acoustic buckling pulses were then modeled us-
ing one envelope per buckling event, with each event repre-
sented by a bipolar, single-pulse waveform. The amplitude
levels, durations and curvatures of the envelope segments
defining these waveforms were based on the acoustic mea-
surements presented in [7] in Figures 3, 4, and 7, and Table
2.

During normal song by C. australasiae, there are 2 or 3
whole cycles of the tymbal plate resonant wave (see Sec-
tion 2.4) in between the buckling pulses (see e.g. Figure 5a
in [7]). In the algorithm, the overall time between the start
of consecutive buckling events already was to be fixed as
the reciprocal of the currently intended musical pitch, as
indicated via the freq_hz input signal. Within this con-
text, the duration of each buckling pulse was then scaled so
as to allow for precisely 2 in-phase echoes before the start
of the next pulse.

Regarding the pulses’ amplitudes, the overall loudness
of tymbal rib buckling events may be subject to further
anatomical control: In [11], it is noted for other cicada
species that contraction of a tensor muscle “increases the
convexity of the tymbal and therefore the power required
to make it click”. This then causes the tymbal ribs to pro-
duce a louder sound once they do buckle. In the algorithm,
the effect of this is mimicked by linear amplitude scaling,
which was made subject to live (musical) control via the
lower half of the input range of loudness_nrm.

Figure 2 shows an example of a complete sequence of
tightly synchronized tymbal rib buckling pulses as com-
puted for pitch E5 by the algorithm.

2.3.2 Modeling Decoherence

So far, we have already recapitulated a crucial part of how
it is that the anatomy of a singing cicada can transform
individual tymbal rib clicks into a longer, continuously os-
cillating pressure wave: The tymbal mechanics of muscle
contraction, apodeme pulling, rib buckling, and plate re-
verberation cause a series of acoustic pulses that are tightly
in phase with eachother. By “decoherence”, then, we will
here mean the loss of this phase coherence.

For the algorithm, we will follow the hypothesis that in
H. maculaticollis, decoherence might explain the change
in timbre that is apparent during the final part of a call se-
quence. In spectrograms of the field recordings (for an ex-
ample, see the top of Figure 3) it is visible how, during
the initial and middle call phases, the sound intensities at
frequencies around insect song frequency are distributed
across separate harmonics; while during the final phase,
these intensities are still relatively higher than elsewhere
in the spectrum, but much more evenly distributed. This
can, of course, also be heard.

The literature indicates that, for C. australasiae, a loss of
phase coherence is associated with slower apodeme pulling
speed resulting in greater intervals between rib buckling
events [7]. For the algorithm, it was assumed that a similar
association would exist for H. maculaticollis. As a first
approximation of concretely modeling this, the duration of
intervals between rib buckling events was made inversely
proportional to apodeme pulling speed.

However, the proportional relationships modeled so
far – i.e. between tymbal muscle contraction frequency,
apodeme pulling speed, and rib buckling intervals – would
then not suffice, still yielding clearly separate harmonics
during the final call phase. Therefore, as a first approxima-
tion of modeling the effects of further unknown processes,
at each tymbal muscle contraction, the pre-buckling inter-
val duration added by decoherence was drawn, according
to a uniformly random distribution, from none up to the
proportionally computed amount.

An example of the resulting algorithm output for pitch E5
can be compared to one of the field recordings in the right-
most part of the spectrograms at the top of Figure 3, start-
ing vertically above the mi_trig impulse in the graph
at the bottom. Also, the underlying, internally computed
tymbalRibs_buckling_sig signal can be heard in

of the Appendices.

2.4 The Tymbal Plate: Resonant Vibrations

As is described in the literature on C. australasiae, when
a tymbal rib buckles, this also rapidly moves inward an-
other part of the tymbal organ: the tymbal plate. This
plate is also connected to the dorsal pad, another compo-
nent of the tymbal made of a rubber-like protein called
resilin. The overall mechanical result is that a buckling
event will set the tymbal plate into a series of resonant vi-
brations, which then become damped over time. Acous-
tically, this means that each buckling pulse will cause a
longer waveform, which then decays approximately expo-
nentially. Two properties can then be used to characterize
such a waveform: the frequency of its wave cycles, and
their “quality factor” Q. The latter is a scalar measure that
quantifies the decrease over time of successive peak ampli-
tudes.

To complicate matters, however, each buckling event
modifies the actually resonating mass, and thereby also the
frequency and Q value of tymbal plate resonation.

In the algorithm, separate instances of a resonant band-
pass filter were used to model the different resonations fol-
lowing each type of buckling pulse. Here, the four center




tymbalRibs_buckling_sig

Duration: 6 ms.

Figure 2. Top: example of simulating one complete se-
quence of tightly synchronized tymbal rib buckling events
(at pitch E5). Bottom: the corresponding simulated re-
sponse of the tymbal plate.

frequencies were explicitly computed from the four mean
resonant wave frequencies presented for C. australasiae
in [7]; while being scaled to 3× freq_hz, so that two de-
caying wave cycles would fit in between tightly coherent
buckling pulses, regardless of the currently required mu-
sical pitch. The four corresponding Q values for C. aus-
tralasiae (measured using acoustically unloaded tymbals)
were then also used explicitly, to compute the different
bandwidths of each filter. Here, care was taken to first
halve the Q values, so as to make them representative of
the acoustically loaded tymbals of an intact cicada.

The overall tymbalPlate_resonation_sig sig-
nal was then computed as the weighted sum of the sig-
nal containing the simulated buckling pulses and those
containing the tymbal plate after-resonations. Here, the
weights were set so that the resulting waveform was similar
to the acoustically measured results from literature. This
included the property that the decayed peak amplitude of
the resonant wave cycle occurring directly before the next
in-buckling pulse should still be above 25% relative to the
maximum – acoustically representing a mechanical vibra-
tion amplitude still powerful enough to help start the next
tymbal rib’s in-buckling (thereby supporting coherence).

The results of this are illustrated in Figure 2, which shows
an example tymbal plate resonation waveform caused by
a complete sequence of tightly synchronized tymbal rib
buckling events, as computed by the algorithm. An ex-
ample tymbalPlate_resonation_sig signal, in-
ternally computed for a full cicada call sequence at pitch
E5, can be heard in of the Appendices.

Finally, one speculative addition was made to the compu-
tation of tymbal plate resonation. This was based on the
impression that perceiving increased loudness during the
calls of H. maculaticollis seemed to possibly involve not
only an increase in overall amplitude, but also a relative in-
crease of the intensities at higher frequencies. This seemed
evident, for example, when listening to the field recording
at [5], and then taking two 4-second snippets from its first
and second half, amplifying these to the same peak am-
plitude, and comparing the spectrograms, especially in the
5.0-8.0 kHz range.

This motivated the idea to extend the algorithm with the

simulation of some mechanism that would cause its out-
put, too, to include a relative increase at higher frequencies
when increasing perceived loudness levels. This seemed
desirable not only from a physical modeling perspective,
but also from a musical one: The suspected phenomenon
seemed reminiscent of how increasing the attack force
while playing notes on acoustic musical instruments can
have an expressive impact by not just increasing the ampli-
tude, but also the brightness of the tones heard.

The mechanism then concretely modeled was itself
highly speculative: Its starting point was that, beyond
some level of increasing applied tymbal muscle force, the
maximum displacement of the vibrating tymbal plate may
become limited to some fixed amount, due to its physi-
cal attachment to the rest of the cicada’s body. Acous-
tically, this then might result in the wave peaks of tym-
bal plate resonation becoming progressively flattened as
vibration amplitude increases, thereby yielding more in-
tense overtones at higher frequencies. In the algorithm’s
tymbalPlate_clippingResonation_sig signal,
the effects of the computations implementing this tentative
idea can be observed. However, these effects can also be
avoided, easily and completely, by not using the upper half
of the loudness_nrm input range. Alternatively, the ex-
tra computations can be removed from the source code by
using an also-included single-line alternative.

2.5 The Abdominal Air Sac, Tympana & Opercula:
Helmholtz Resonation

As is described in the literature on C. australasiae, move-
ment of the tymbal plate causes air pressure changes going
into the abdominal air sac. Together with the left and right
tympana – which, variably covered by the opercula, act as
the effective sound sources radiating into the external sur-
roundings – this air sac forms an abdominal resonator that
appears to produce bottle-shaped or Helmholtz resonation,
having the effect of a narrow band-pass filter. Here, the
dominant resonant frequency f 0 is closely tuned to that
of the tymbal, and can be modeled by the equations used
in [8] that are listed in (1).

𝑓 0 =
𝑐

2𝜋
·
√︂

𝐴

𝐿 · 𝑉

𝐴 = 2 · 𝑎

𝐿 = 16 · 𝑟 / 3𝜋

𝑟 =
√︀

𝑎 / 𝜋

(1)

Here, c is the speed of sound; A is the area of the neck,
being twice the area a of a single tympanum; L is the ef-
fective length of the neck, determined by a via the single-
tympanum equivalent hole radius r; and V is the volume of
the cavity. From these dependencies, it then follows that
the main variables remaining to be modeled for the case of
H. maculaticollis are V and A, i.e. abdominal volume and
effective tympanal area.

Waveform & spectrogram: field recording of a complete H. maculaticollis call sequence.        Duration everywhere: 6.6 seconds.

Waveform & spectrogram: example of corresponding algorithm output computed for pitch E5.

dB

Hz

abdominal_resonation_sig

Waveforms of selected, corresponding internal signals:

opercular_attenuation

c_austral_tympanal_area_m2

c_austral_abdominal_volume_m3

tymbalApodeme_pullingSpeed_nrm

tymbalMuscleFreq_hz

Waveform of corresponding real-time input signals:

mae_trig mae_retrig mae_retrig mae_retrig mi_trig

dB

1.0 ~ 100 × 10-6 m2 

0.5 ~ 5.0 × 10-6  m3 

0.20 ~ 200 Hz

Figure 3. Algorithmically simulating the calls of H. maculaticollis. Each signal is further explained in the main text.



In the algorithm, abdominal resonation was modeled us-
ing a 6-pole (i.e. -36 dB/octave) resonant bandpass filter.
Its bandwidth was tuned so as to make algorithm output
harmonically similar to H. maculaticollis field recordings,
using spectrum plot measurements of the progressive drop
in amplitude of the harmonic peaks below insect song fre-
quency (details are in the source code). Fortunately, mu-
sically, there was no actual trade-off here when testing at
pitch E5, as the resulting bandwidth value seemed to fall
within a middle range only outside of which the unintended
perception of pitches E6/B6 and E4 seemed to start to oc-
cur.

Then, for computing the abdominal volume over time,
the starting point were measurements of unextended and
extended abdominal cavity volume for C. australasiae, de-
rived from [8]. To this was then added the general idea
that H. maculaticollis’ abdominal cavity might be less ex-
tended during its final call phase. Once modeled, the ef-
fects of this seemed hard to discern, however, both via
auditory perception and in spectrograms, and the inter-
ested reader is therefore referred to the source code for
further motivation and details. The resulting computa-
tion of c_austral_abdominal_volume_m3 is illus-
trated by the waveform placed fourth-lowest in Figure 3.

In comparison, modeling H. maculaticollis’ effective
tympanal area over time was considerably more involved,
once similarly started from the overall tympanal area re-
ported for C. australasiae in [8]. In general, modeling was
based on the observation that both cicada species raise their
abdomen in some way while singing, thereby lifting the
tympana away from the opercula which cover them; 3 and
also, on the observation in spectrograms of the H. macu-
laticollis field recordings that during the initial and mid-
dle call phases, this modulation of effective tympanal area
seemed to be reliably synchronized with specific patterns
in tymbal muscle contraction frequency.

As for tymbal muscle frequency in Section 2.1, a system
of envelopes was implemented, one for each insect call
phase, with the specific target levels, durations and cur-
vatures defined by a total of 26 scalar constants explicitly
based on measurements of amplitude and spectral features
present in the field recording obtained from [4]. During
this, at three points in the algorithm, musical requirements
were consciously prioritized over realistic physical mod-
eling. In each case, envelope segment properties were ad-
justed so as to perceptually make each simulated insect call
phase better correspond to the perceived attack of a single
tonal event. This affected 6 constants. 4 The details of all
of this are documented in the source code.

A fourth point where musical requirements were priori-
tized over realistic physical modeling was when recomput-
ing the abdominal resonation frequency resulting from the
above relative to the currently intended musical pitch (as

3 For an H. maculaticollis video example, see:
https://youtu.be/XFOaGPNT55U?t=157.

4 c_MAETRIG_TYMPAREA_START_NRM,
c_MAETRIG_TYMPAREA_RISE_DUR_S,
c_MAERETRIG_TYMPAREA_START_NRM,
c_MAERETRIG_TYMPAREA_RISE_DUR_S,
c_MITRIG_TYMPAREA_MAX_NRM,
c_MITRIG_TYMPAREA_END_NRM.

specified via freq_hz). This, however, will be discussed
in the Discussion section.

The resulting computation of c_austral_
tympanal_area_m2 is illustrated by the wave-
form placed fifth-lowest in Figure 3. Also, an ex-
ample of the overall resulting unattenuated_
abdominal_resonation_sig signal can be heard in

of the Appendices, as it was internally
computed for a full cicada call sequence at pitch E5.

2.6 The Opercula: Amplitude Attenuation

In the literature on C. australasiae, a reduction in sound
pressure level is reported when fully closing the oper-
cula. Accordingly, in the algorithm, too, amplitude at-
tenuation of the abdominal resonation signal was imple-
mented, based on the current effective tympanal area.
This was done down to a dB SPL level set so that al-
gorithm output matched the drop in strongest harmonic
peak amplitude observed during middle call phases in
the H. maculaticollis field recording at [4]. An exam-
ple of the resulting opercular_attenuation sig-
nal is illustrated by the waveform placed sixth-lowest
in Figure 3, as it was internally computed for a full
cicada call sequence at pitch E5. The corresponding
abdominal_resonation_sig signal can be visually
compared to the field recording using the spectrograms and
waveforms aligned at the top of Figure 3, while it can be
heard in of the Appendices.

3. DISCUSSION

From the literature on C. australasiae, it is clear that its
abdominal resonation – when abdominal volume is fully
extended, and tympanal area fully uncovered – is strongest
around the same frequency as tymbal plate resonation. Ac-
cordingly, in the algorithm, the abdominal resonation fre-
quency initially was scaled so as to match tymbal plate res-
onation frequency. This meant that it, too, followed the
currently intended musical pitch.

However, when testing this at pitch E4 instead of E5,
output from abdominal resonation perceptually seemed to
change not only in pitch, but also in timbre, and this to such
an extent as to suggest origination from another, different
sound source. This seemed problematic, in hindering algo-
rithm output from being perceived as coming from a single
source, e.g. during melodic use with other musical voices.
An example of this pitch-following abdominal resonation
can be heard in of the Appendices,
which contains the abdominal_resonation_sig
signal for a full call sequence at pitch E4.

Scaling abdominal resonation frequency relative to a
fixed pitch E5 then seemed to perform better in this re-
gard. In , the corresponding example
of fixed-pitch abdominal resonation can be heard. Both
options are preserved in the source code.

4. CONCLUSION

In Sections 2 and 3, we have discussed the design and im-
plementation of a new algorithm that can simulate the calls





https://youtu.be/XFOaGPNT55U?t=157


of H. maculaticollis, subject to real-time control. Now,
revisiting this, as a case study of the possible trade-offs
when modeling bioacoustics for tonal use, we can identify
11 instances, extending from the input parameters to the
pre-final abdominal resonation stage, where a musical re-
quirement was consciously prioritized over faithful physi-
cal modeling.

This was done for the soprano freq_hz input range;
the pitch-inducing number of buckling events per muscle
contraction; the whole-octave drop in tymbal muscle fre-
quency; the harmonics kept in tune across repeated middle
call phases; the pitch-adjusted tymbal plate resonation fre-
quency; the optionally increasing timbral brightness dur-
ing increasing loudness; the pitch-respecting abdominal
resonation bandwidth; the three tympanal area envelope
details for better matching insect call phases with single-
attack tonal events; and the fixed-pitch abdominal res-
onation frequency.

This seems to illustrate a number of points that may be
relevant when modeling bioacoustics for tonal use. First,
that it may be necessary to prioritize musical requirements
over faithful physical modeling at many stages of design
and implementation. Second, that the resulting adjust-
ments may revolve around having physical modeling per-
ceptually yield sonic events that are well-pitched, single-
attack, single-source, and timbrally expressive. Third,
that an inherent contradiction may be encountered (here,
while modeling abdominal resonation, see Section 3): The
pitch-adjusted simulation of resonating bodies may pro-
duce sonic results that fail musically precisely when they
succeed physically, i.e. when inducing the perception of
different sound sources at different pitches.

Finally, the algorithm itself is provided in the Appen-
dices of this paper, together with the audio illustrations
that have already been used to demonstrate the computa-
tion of some of its key internal signals. In addition to this,

demonstrates the tonal use of overall
algorithm output together with another instrument, con-
taining brief examples of counterpoint and chordal accom-
paniment by a digital piano. Throughout the algorithm’s
source code, emphasis has been put on readability and ease
of reuse: The overall structure is flat and unambiguous, and
uses human-readable constants and variables together with
detailed in-line comments that explicitly motivate each of
the choices made. The intention here was to make it eas-
ier for (in principle) anybody to use, modify or extend the
algorithm. This could include changes to more faithfully
simulate the calls of cicadas in general (e.g., by explicitly
modeling a pair of tymbals); changes to better match H.
maculaticollis, C. australasiae, or any other species of ci-
cada; and changes to the trade-offs between realistic phys-
ical modeling and musical use.

5. APPENDICES

The appendices are included inside the PDF of this paper.
Each embedded file can be opened or exported using its
(shortened) name in the text. For the Audio Illustrations,
this is via: , , , , , ,

, and . For the source code of the algorithm, this is
via: . The lat-
ter needs to be placed among the Extensions of the Su-
perCollider installation that is then used to run the code
(details provided). Finally, example source code that in-
stantiates and parametrizes the algorithm is provided via:

.

/* This is SuperCollider code, written by Staas de Jong.

   SuperCollider Extension containing the \maemi algorithm.

   NB: To run the algorithm, this file ("SoundGeneratingProcesses.sc") should be placed in
       the Extensions directory of your SuperCollider installation, inside two subdirectories,
       as in:

       "YOUR_EXTENSIONS_DIRECTORY\SoundGeneratingProcesses\classes\SoundGeneratingProcesses.sc"
*/

SoundGeneratingProcesses
/* container class for the dynamic instantiation of sound-generating processes
*/
{
  // private:
  var audioSynthesis_group = nil,
      stereoOut_buses = nil,
      stereoRec_buses = nil,
      stereoOutAtt_nrm = nil,
      stereoRecAtt_nrm = nil,
      audioStereoOutAndRec_func = nil;


  *new
  /* constructor */
  {
    arg audioSynthesis_group = nil,
        stereoOut_buses = [12, 13], // custom default
        stereoOutAtt_db = -8,
        stereoRec_buses = nil,
        stereoRecAtt_db = nil;

    if (audioSynthesis_group == nil)
    { Error ("SoundGeneratingProcesses: audioSynthesis_group not specified.").throw;
    };

    ^super.new.soundGeneratingProcesses_init (audioSynthesis_group, stereoOut_buses, stereoOutAtt_db,
                                                                    stereoRec_buses, stereoRecAtt_db);
  }


  soundGeneratingProcesses_init
  /* initialization (of a new object) */
  {
    arg audioSynthesis_group_arg, stereoOut_buses_arg, stereoOutAtt_db_arg,
                                  stereoRec_buses_arg, stereoRecAtt_db_arg;

    audioSynthesis_group = audioSynthesis_group_arg;
    stereoOut_buses = stereoOut_buses_arg;
    stereoRec_buses = stereoRec_buses_arg;
    if (stereoOutAtt_db_arg != nil)
    { stereoOutAtt_nrm = stereoOutAtt_db_arg.dbamp;
    };
    if (stereoRecAtt_db_arg != nil)
    { stereoRecAtt_nrm = stereoRecAtt_db_arg.dbamp;
    };


    // define the function, to be used by all sound-generating algorithms below,
    // that implements attenuated stereo output to the transducer and/or recording buses
    audioStereoOutAndRec_func =
    { arg left_sig = nil, right_sig = nil;

      if (stereoOut_buses != nil)
      { Out.ar (stereoOut_buses[0], stereoOutAtt_nrm * left_sig);
        Out.ar (stereoOut_buses[1], stereoOutAtt_nrm * right_sig);
      };

      if (stereoRec_buses != nil)
      { Out.ar (stereoRec_buses[0], stereoRecAtt_nrm * left_sig);
        Out.ar (stereoRec_buses[1], stereoRecAtt_nrm * right_sig);
      };
    };


    // Compile the "maemi" algorithm, which aims to
    // (1) mimick the calls of H. maculaticollis, a cicada native to the Korean Peninsula;
    // (2) with such calls, however, appearing to have a controllable pitch, from E4 to E5;
    // (3) while also, across control input, still seeming to come from a single heard source.
    //
    // The algorithm uses physical modeling, based on results from literature and reference
    // recordings. This is adapted, however, to respect the intended perceived pitch.
    //
    // Some of the control parameters' names are based on "maemi", the onomatopoeic Korean
    // word for cicada, to indicate which phase of the insect call they refer to.
    SynthDef
    ( name: \maemi,
      ugenGraphFunc: // NB: outputs *only* via audioStereoOutAndRec_func()
      {
        arg a_gate_bit   = 1.0,             // function: on 0.0: output completes, synth then freed
                                            // verified: .set messages also yield uninterpolated
                                            // audio-rate step changes
            loudness_nrm = 0.5,             // normalized scale across which perceived loudness increases:
                                            // - from 0.0 to 0.5: due to amplitude scaling
                                            // - from 0.5 to 1.0: due to changing harmonic content
            freq_hz      = 659.73703832729, // default: pitch E5, Railsback-corrected
            t_mae_trig   = 0.0,             // assumption: the triggers...
            t_mae_retrig = 0.0,             //             ...are never...
            t_mi_trig    = 0.0;             //             ...used simultaneously


        // variables used as general constants
        var c_RAILSBACK_OCT5_E_HZ = 659.73703832729, // pitches E5 & E4, Railsback-corrected
            c_RAILSBACK_OCT4_E_HZ = 329.59373351773;
        var c_MAEMI_ENVS_DURSCALE = 1.0; // general time scaling factor for envelopes


        // convert to audio rate: the input signals triggering different phases of the insect call
        var a_mae_trig   = T2A.ar (t_mae_trig);   // the initial phase
        var a_mae_retrig = T2A.ar (t_mae_retrig); // the middle phase(s)
        var a_mi_trig    = T2A.ar (t_mi_trig);    // the final phase


        // model tymbal muscle contractions
        //
        // in general, the frequency of these:
        // - is modeled as a normalized, linear value from 0.0 to around 1.0, where 1.0 produces
        //   synchronized buckling that yields an overall waveform inducing the perceived pitch
        //   specified by freq_hz
        // - seems to alternate, in H. maculaticollis reference audio recordings, between values
        //   near 1.0 during the initial & middle phases; and lower values during the final phase
        //
        // a musical choice is then:
        // - to have the lower tymbal muscle frequencies be near a lower, whole octave:
        //   - harmonically, so as to clearly connect to the initially heard pitch
        //   - harmonically, so as to help avoid unintended dissonances during use with
        //     other musical voices in a piece
        //
        // physical modeling choices are then:
        // - to go 1 octave lower: 2 seemed to sound similar to a general end of cicada calling
        // - to use envelopes for the (more detailed) control of tymbal muscle frequency, with
        //   separate sections dynamically triggered for each insect call phase
        // - to base the contents of these envelopes on an attempt to have algorithm output mimick
        //   an H. maculaticollis reference recording,
        //   - comparing 12 harmonics evident in spectrograms sampled at 11025 Hz - visible in the
        //     3.3-5.3 kHz range of the reference, and in the 0.9-2.7 kHz range of algorithm output
        //
        // a musical choice is then:
        // - to *not* mimick a downward trend in tymbal muscle frequency that seemed evident across
        //   the repeated middle phases: this to help avoid going out of tune during tonal use
        //
        var c_MAETRIG_TYMBALMUSCLEFREQ_STARTSEGMENT_DUR_S = 0.130 * c_MAEMI_ENVS_DURSCALE;
        var c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S      = 0.065 * c_MAEMI_ENVS_DURSCALE;
        var c_MAETRIG_TYMBALMUSCLEFREQ_LINRISE_DUR_S      = 0.525 * c_MAEMI_ENVS_DURSCALE;
        var c_MAETRIG_TYMBALMUSCLEFREQ_LINFALL_DUR_S      = 0.385 * c_MAEMI_ENVS_DURSCALE;
        var c_MAETRIG_TYMBALMUSCLEFREQ_LINRISE_START_NRM  = 1.00 - 0.119;
        var c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM = 0.0225;
        var c_MAETRIG_TYMBALMUSCLEFREQ_LINFALL_START_NRM  = 1.00 + 0.040;
        //
        var tymbalMuscleFreq_maeTrig_mulEnv = Env.new
        ( [ 0.0,
            // target levels
            0.5,
            c_MAETRIG_TYMBALMUSCLEFREQ_LINRISE_START_NRM + c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM,
            c_MAETRIG_TYMBALMUSCLEFREQ_LINRISE_START_NRM,
            c_MAETRIG_TYMBALMUSCLEFREQ_LINFALL_START_NRM,
            c_MAETRIG_TYMBALMUSCLEFREQ_LINFALL_START_NRM + c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM,
            c_MAETRIG_TYMBALMUSCLEFREQ_LINFALL_START_NRM,
            1.00
          ],
          [ // durations
            0.0,
            c_MAETRIG_TYMBALMUSCLEFREQ_STARTSEGMENT_DUR_S + (c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5),
            c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5,
            c_MAETRIG_TYMBALMUSCLEFREQ_LINRISE_DUR_S,
            c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5,
            c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5,
            c_MAETRIG_TYMBALMUSCLEFREQ_LINFALL_DUR_S
          ],
          [ // curvatures
            0.0,
            \exponential, // here, listening, a faster-than-linear rise seemed apparent
            \linear,
            \linear,
            \linear,
            \linear,
            \linear
          ]
        );
        var tymbalMuscleFreq_maeTrig_mulEnv_sig = EnvGen.ar
        ( envelope: tymbalMuscleFreq_maeTrig_mulEnv,
          gate:     a_mae_trig
                    - a_mae_retrig // "on this trigger...
                    - a_mi_trig    //  ...go to final envelope level, immediately"
        );
        //
        var c_MITRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S
            = c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S;
        var c_MITRIG_TYMBALMUSCLEFREQ_NONLINFALL_DUR_S
            = 0.44 * c_MAEMI_ENVS_DURSCALE;
        var c_MITRIG_TYMBALMUSCLEFREQ_KEEPLEVEL_DUR_S
            = 999.0;
        var c_MITRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM
            = c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM;
        //
        var tymbalMuscleFreq_miTrig_mulEnv = Env.new
        ( [ 0.0,
            // target levels
            1.0,
            1.0 + c_MITRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM,
            0.5,
            0.5,
            1.0
          ],
          [ // durations
            0.0,
            c_MITRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5,
            c_MITRIG_TYMBALMUSCLEFREQ_NONLINFALL_DUR_S,
            c_MITRIG_TYMBALMUSCLEFREQ_KEEPLEVEL_DUR_S,
            0.0
          ],
          [ // curvatures
            0.0,
            \linear,
            \sine,
            \linear,
            0.0
          ]
        );
        var tymbalMuscleFreq_miTrig_mulEnv_sig = EnvGen.ar
        ( envelope: tymbalMuscleFreq_miTrig_mulEnv,
          gate:     a_mi_trig
                    - a_mae_trig   // "on this trigger...
                    - a_mae_retrig //  ...go to final envelope level, immediately"
        );
        //
        var c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S
            = c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S;
        var c_MAERETRIG_TYMBALMUSCLEFREQ_LINRISE_DUR_S
            = 0.176 * c_MAEMI_ENVS_DURSCALE;
        var c_MAERETRIG_TYMBALMUSCLEFREQ_LINFALL_DUR_S
            = 0.285 * c_MAEMI_ENVS_DURSCALE;
        var c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM
            = c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM;
        //
        var tymbalMuscleFreq_maeRetrig_mulEnv = Env.new
        ( [ 0.0,
            // target levels
            1.0,
            1.0 + c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM,
            1.0,
            1.0 + c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM,
            1.0,
            1.0 + c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_HEIGHT_NRM,
            1.0
          ],
          [ // durations
            0.0,
            c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5,
            c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5,
            c_MAERETRIG_TYMBALMUSCLEFREQ_LINRISE_DUR_S,
            c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5,
            c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5,
            c_MAERETRIG_TYMBALMUSCLEFREQ_LINFALL_DUR_S
          ],
          [ // curvatures
            0.0,
            \linear,
            \linear,
            \linear,
            \linear,
            \linear,
            \linear
          ]
        );
        var tymbalMuscleFreq_maeRetrig_mulEnv_sig = EnvGen.ar
        ( envelope: tymbalMuscleFreq_maeRetrig_mulEnv,
          gate:     a_mae_retrig
                    - a_mae_trig // "on this trigger...
                    - a_mi_trig  //  ...go to final envelope level, immediately"
        );
        //
        var tymbalMuscleFreq_mulEnv_sig
            = tymbalMuscleFreq_maeTrig_mulEnv_sig
            * tymbalMuscleFreq_maeRetrig_mulEnv_sig
            * tymbalMuscleFreq_miTrig_mulEnv_sig;
        var tymbalMuscleFreq_hz = (freq_hz / 4.0) // base musical pitch on at most 3 in-buckling pulses
                                                  // followed by 1 out-buckling pulse
                                  * tymbalMuscleFreq_mulEnv_sig;
        //
        // model a process introducing random changes in the tymbal muscle contraction frequency,
        // with specific properties that are (highly) speculative:
        // - type: the controlling effect is subject to many (unknown, but) smaller additive effects:
        //   - semi-normally distributed random values are added/subtracted to/from the frequency
        //   - musically, this approach helps preserve an accurate average pitch
        // - update rate: at least once every tymbal muscle contraction, up to those yielding E5 pitch
        // - peak amplitude: 10 cents
        //   - perceptually, this seems to yield at most a slight variation
        //   - going higher, however, resulted in ripples in muscle frequency harmonics not present in
        //     reference recordings:
        //     - comparing 12 harmonics evident in spectrograms sampled at 11025 Hz - visible in the
        //       3.3-5.3 kHz range of the reference, and in the 0.9-2.7 kHz range of algorithm output
        //
        var c_TYMBALMUSCLEFREQ_RND_UPDATE_RATE_HZ
            = c_RAILSBACK_OCT5_E_HZ / 4.0;
        var c_TYMBALMUSCLEFREQ_RND_UPDATE_MAX_AMPL_CENTS // a bipolar amplitude
            = 10.0;
        var c_ONE_CENT_FACTOR
            = 1.0005777895065549;
        //
        var tymbalMuscleFreq_rndUpdate_trig
            = Impulse.kr (c_TYMBALMUSCLEFREQ_RND_UPDATE_RATE_HZ);
        var tymbalMuscleFreq_rndUpdate_maxAmpl_nrm
            = DC.kr ((c_ONE_CENT_FACTOR ** c_TYMBALMUSCLEFREQ_RND_UPDATE_MAX_AMPL_CENTS) - 1.0);
        var tymbalMuscleFreq_rndUpdate_factor
            = Latch.kr
              ( in:   tymbalMuscleFreq_rndUpdate_maxAmpl_nrm.sum3rand,
                trig: tymbalMuscleFreq_rndUpdate_trig
              ) + 1.0;
        var randomizedTymbalMuscleFreq_hz
            = tymbalMuscleFreq_hz * tymbalMuscleFreq_rndUpdate_factor;
        //
        // model tymbal muscle contractions:
        // - as a unipolar waveform consisting of single-sample impulses
        var tymbalMuscle_trig = Impulse.ar
        ( freq: randomizedTymbalMuscleFreq_hz
        );


        // model the separate buckling pulse waves of the tymbal ribs
        var c_TYMBAL_RIBS_BUCKLING_ATT_NRM  = 0.5; // output amplitude attenuation
        var tymbalRib1_buckleEnv_levels     = [ 0.0,  0.0,  1.0/2.3,     -1.0,      0.0 ]
                                              * c_TYMBAL_RIBS_BUCKLING_ATT_NRM;
        var tymbalRib1_buckleEnv_durations  = [         0,    0.225,      0.5,    0.275 ];
        var tymbalRib1_buckleEnv_curvatures = [         0,    \sine,  \linear,  \linear ];
        var tymbalRib1_buckleEnv = Env.new // the basic amplitude envelope to derive buckling pulses from
        ( tymbalRib1_buckleEnv_levels,     // positive amplitude ~ inward pressure into abdominal resonator
          tymbalRib1_buckleEnv_durations,
          tymbalRib1_buckleEnv_curvatures
        );
        var tymbalRib2_buckleEnv = Env.new
        ( tymbalRib1_buckleEnv_levels * 1.01,
          tymbalRib1_buckleEnv_durations,
          tymbalRib1_buckleEnv_curvatures
        );
        var tymbalRib3_buckleEnv = Env.new
        ( tymbalRib1_buckleEnv_levels * 0.83,
          tymbalRib1_buckleEnv_durations,
          tymbalRib1_buckleEnv_curvatures
        );
        var tymbalRibs_outBuckleEnv = Env.new
        ( tymbalRib1_buckleEnv_levels * -0.34,
          tymbalRib1_buckleEnv_durations,
          tymbalRib1_buckleEnv_curvatures
        );


        // model the effect of loudness control at the tymbal, which is described in
        // literature for other species of cicada:
        // - contraction of a tensor muscle "increases the convexity of the tymbal and
        //   therefore the power required to make it click", causing the tymbal ribs to
        //   produce a louder sound once they buckle
        // - the effect of this is mimicked here by linear amplitude scaling, which
        //   happens under the (musical) control of an external loudness parameter
        var tymbalRibs_buckling_att = loudness_nrm.linlin (0.0, 0.5, 0.0, 1.0, clip: 'minmax');


        // model tymbal rib buckling, including decoherence:
        //
        // - from literature: for tymbal rib buckling events by C. australasiae, their phase coherence as a
        //   pulse train is supported by the mechanics of the insect's anatomy, from tymbal apodeme to abdominal
        //   air sac [Bennet-Clark & Young 1992]
        // => by "decoherence", we here mean the loss of this phase coherence
        //
        // - from literature: for C. australasiae, a loss of phase coherence is associated with slower apodeme
        //   pulling speed resulting in greater intervals between rib buckling events [Bennet-Clark 1997]
        // => here, we assume a similar association exists for H. maculaticollis
        //    ^ concretely, as a first approximation: that the duration of intervals between rib buckling events
        //      is inversely proportional to apodeme pulling speed
        // => additionally, we also assume that apodeme pulling speed is related to tymbal muscle contraction
        //    frequency
        //    ^ concretely, as a first approximation: that it is directly proportional to this frequency
        //
        // - from literature: at higher apodeme pulling speeds: "It appears that the tymbal buckling mechanism
        //   offers good coupling between the vibrations of the ribs so that the rapidly excited tymbal
        //   inevitably produces a coherent waveform" [Bennet-Clark 1997]
        // => interpretation: above a certain threshold of apodeme pulling speed, tymbal mechanics will pull
        //    buckling into (the same) tight coherence
        //    ^ implementation: apodeme pulling speed is clipped and scaled to a maximum corresponding to this
        //      threshold
        //      ^ the threshold value is then chosen so that relatively little decoherence occurs during the
        //        initial and middle phases of the cicada call, i.e. where tymbal muscle contraction frequency
        //        seemed to be relatively higher based on reference audio recordings of H. maculaticollis
        //
        // - observation from such reference recordings: during the final phase of the insect's call, it can be
        //   heard & is visible in spectrograms that there are still relatively higher sound intensities at
        //   frequencies around insect song frequency; but also, that these intensities are now much more evenly
        //   distributed than across the separate harmonics visible during the initial and middle phases
        // => hypothesis we will follow: decoherence may explain this change in timbre during the final phase of
        //    the call of H. maculaticollis
        //    - however, the proportionalities modeled so far (between tymbal muscle contraction frequency,
        //      apodeme pulling speed, and rib buckling intervals) then do not suffice, still yielding clearly
        //      separate harmonics
        //    - therefore, as a first approximation to model the effects of further unknown processes, at each
        //      tymbal muscle contraction, the interval duration added by decoherence is drawn from none up to
        //      the proportionally computed amount, according to a uniformly random distribution
        var interBucklingDelay_s
            = DC.ar (1.0) / freq_hz;
        var tymbalApodeme_pullingSpeed_nrm
            = Clip.ar (in: tymbalMuscleFreq_mulEnv_sig, lo: 1e-10, hi: 0.9) / 0.9;
        var decoherence_maxDelayScaling
            = DC.ar (1.0) / tymbalApodeme_pullingSpeed_nrm;
        var perBuckle_maxAddedDecoherenceInterval_nrm
            = decoherence_maxDelayScaling - DC.ar (1.0);
        var perBuckle_maxAddedDecoherenceInterval_s
            = interBucklingDelay_s * perBuckle_maxAddedDecoherenceInterval_nrm;
        var perBuckle_addedDecoherenceInterval_s
            = Latch.ar
              ( in:   perBuckle_maxAddedDecoherenceInterval_s.rand,
                trig: tymbalMuscle_trig
              );
        //
        // model the in-buckling of tymbal ribs 1-3 on muscle contraction
        var tymbalRibs_buckleDur_s = interBucklingDelay_s // each in-buckling pulse can have...
                                     / 3.0; // ...2 in-phase echoes before the next one starts
        var tymbalRib1_inBuckle_sig = EnvGen.ar
        ( envelope:  tymbalRib1_buckleEnv,
          gate:      DelayC.ar
                     ( in:           tymbalMuscle_trig,
                       maxdelaytime: 1.0,
                       delaytime:    (1.0 * perBuckle_addedDecoherenceInterval_s)
                     ),
          timeScale: tymbalRibs_buckleDur_s
        ) * tymbalRibs_buckling_att;
        var tymbalRib2_inBuckle_sig = EnvGen.ar
        ( envelope:  tymbalRib2_buckleEnv,
          gate:      DelayC.ar
                     ( in:           tymbalMuscle_trig,
                       maxdelaytime: 1.0,
                       delaytime:    (2.0 * perBuckle_addedDecoherenceInterval_s) +
                                     (1.0 * interBucklingDelay_s)
                     ),
          timeScale: tymbalRibs_buckleDur_s
        ) * tymbalRibs_buckling_att;
        var tymbalRib3_inBuckle_sig = EnvGen.ar
        ( envelope:  tymbalRib3_buckleEnv,
          gate:      DelayC.ar
                     ( in:           tymbalMuscle_trig,
                       maxdelaytime: 1.0,
                       delaytime:    (3.0 * perBuckle_addedDecoherenceInterval_s) +
                                     (2.0 * interBucklingDelay_s)
                     ),
          timeScale: tymbalRibs_buckleDur_s
        ) * tymbalRibs_buckling_att;
        //
        // model the out-buckling of tymbal ribs 1-3 on muscle contraction
        var tymbalRibs_outBuckle_sig = EnvGen.ar
        ( envelope:  tymbalRibs_outBuckleEnv,
          gate:      DelayC.ar
                     ( in:           tymbalMuscle_trig,
                       maxdelaytime: 1.0,
                       delaytime:    (4.0 * perBuckle_addedDecoherenceInterval_s) +
                                     (3.0 * interBucklingDelay_s)
                     ),
          timeScale: tymbalRibs_buckleDur_s
        ) * tymbalRibs_buckling_att;


        // model the complete series of tymbal rib buckling pulses on muscle contraction
        var tymbalRibs_buckling_sig = tymbalRib1_inBuckle_sig +
                                      tymbalRib2_inBuckle_sig +
                                      tymbalRib3_inBuckle_sig +
                                      tymbalRibs_outBuckle_sig;


        // model the response of the tymbal plate to the tymbal rib buckling pulses:
        // (1) lowpass-filter the buckling pulses to avoid overly abrupt amplitude changes
        var tymbalRibs_buckling_smoothed_sig = RLPF.ar
        ( in:   tymbalRibs_buckling_sig,
          freq: 10000,
          rq:   1.0
        );
        // (2) based on results from literature, model the tymbal plate's resonation:
        // - each buckling pulse causes a longer, approx. exponentially decaying waveform
        // - in general, such resonation can be simulated using some bi-quadratic digital
        //   filter of less than 6 poles
        // - during rapid apodeme pulls (i.e., cicada song) the intervals between in-buckling
        //   pulses become short & phase-locked
        // - this is for tymbal plate mechanical reasons: previous after-vibration has
        //   > 0.25 peak amplitude? helps start the next rib's in-buckling
        // - in this way, 2 or 3 whole cycles of the resonant wave fit between in-buckling pulses
        // - in-buckling ribs modify the resonating mass, however, so that the frequency of the
        //   resonant wave is adjusted as ribs buckle in
        // additionally found, by trial and error:
        // - modeling this using a single BBandPass.ar, frequency-modulated by an envelope:
        //   - did work at a buckling frequency around 330 Hz
        //   - did not work around 660 Hz: wrong # resonation cycles after first 4 buckling pulses
        // -> approach: give each rib buckling its own, statically parametrized BBandPass.ar
        //
        // variables used as constants:
        // resonation after different buckling events has different amounts of damping,
        //   (with *higher* 'bw' values (see below) apparently yielding higher damping)
        // which are computed in a way that reflects results from literature
        //   (where *lower* 'Q' values apparently represent higher damping)
        var c_RIB1_IN_DAMPING_BW  = (10.0 / 2.0).linlin (inMin: 3.0, inMax: 5.5, outMin: 0.5, outMax: 0.25);
        var c_RIB2_IN_DAMPING_BW  = (10.5 / 2.0).linlin (inMin: 3.0, inMax: 5.5, outMin: 0.5, outMax: 0.25);
        var c_RIB3_IN_DAMPING_BW  = (11.0 / 2.0).linlin (inMin: 3.0, inMax: 5.5, outMin: 0.5, outMax: 0.25);
        var c_RIBS_OUT_DAMPING_BW = ( 6.0 / 2.0).linlin (inMin: 3.0, inMax: 5.5, outMin: 0.5, outMax: 0.25);
        // model the after-resonations separately for each type of buckling event
        var tymbalPlate_rib1_in_afterResonations_sig = BBandPass.ar
        ( tymbalRib1_inBuckle_sig,
          freq: freq_hz * 3.0 *    // choice: create 2 decaying wave cycles between buckling pulses,
                (4370.0 / 4370.0), // frequency-adjusted due to buckling (result from literature)
          bw:   c_RIB1_IN_DAMPING_BW
        );
        var tymbalPlate_rib2_in_afterResonations_sig = BBandPass.ar
        ( tymbalRib2_inBuckle_sig,
          freq: freq_hz * 3.0 *    // choice: create 2 decaying wave cycles between buckling pulses,
                (4190.0 / 4370.0), // frequency-adjusted due to buckling (result from literature)
          bw:   c_RIB2_IN_DAMPING_BW
        );
        var tymbalPlate_rib3_in_afterResonations_sig = BBandPass.ar
        ( tymbalRib3_inBuckle_sig,
          freq: freq_hz * 3.0 *    // choice: create 2 decaying wave cycles between buckling pulses,
                (3920.0 / 4370.0), // frequency-adjusted due to buckling (result from literature)
          bw:   c_RIB3_IN_DAMPING_BW
        );
        var tymbalPlate_ribs_out_afterResonations_sig = BBandPass.ar
        ( tymbalRibs_outBuckle_sig,
          freq: freq_hz * 3.0 *    // choice: create 2 decaying wave cycles between buckling pulses,
                (6540.0 / 4370.0), // frequency-adjusted due to buckling (result from literature)
          bw:   c_RIBS_OUT_DAMPING_BW
        );
        var tymbalPlate_afterResonations_sig = tymbalPlate_rib1_in_afterResonations_sig +
                                               tymbalPlate_rib2_in_afterResonations_sig +
                                               tymbalPlate_rib3_in_afterResonations_sig +
                                               tymbalPlate_ribs_out_afterResonations_sig;
        // (3) add tymbalPlate_afterResonations_sig to tymbalRibs_buckling_smoothed_sig,
        //     attenuating each so that the resulting waveform is similar to results from
        //     literature, incl. wave cycle amplitudes > 0.25 before a subsequent in-buckling pulse
        var tymbalPlate_resonation_sig = (0.5 * tymbalRibs_buckling_smoothed_sig) +
                                         (1.5 * tymbalPlate_afterResonations_sig);


        // model a mechanism whereby increasing perceived loudness during cicada calls
        // can also include a relative increase in intensity at higher frequencies:
        // - this seems to occur in reference recordings
        // - the mechanism modeled below is then based on the idea that presumably,
        //   beyond some level of increasingly applied tymbal muscle force, the tymbal
        //   plate's vibrational peak amplitude cannot increase any further, due to the
        //   plate's attachment to the rest of the insect's body
        // - this might mean that vibrational amplitude that otherwise would have been
        //   greater stays level at some maximum possible amplitude
        // - both this proposed mechanism and its implementation are highly speculative,
        //   and can be easily removed by replacing the lines below by:
        //   "var tymbalPlate_clippingResonation_sig = tymbalPlate_resonation_sig;"
        var c_TYMBALPLATE_CLIPPING_AMPLITUDE
            = 0.415; // determined empirically: just outside tymbalPlate_resonation_sig's max. peak amplitude
        var c_TYMBALPLATE_AMPLIFICATION_MAX
            = 2.0; // chosen subjectively for yielding a clear maximum timbral difference, linearly scaling
                   // toward which still seemed to perceptually happen relatively evenly
        var tymbalPlate_amplification
            = loudness_nrm.linlin (0.5, 1.0, 1.0, c_TYMBALPLATE_AMPLIFICATION_MAX, clip: 'minmax');
        var tymbalPlate_clippingResonation_sig
            = ( tymbalPlate_resonation_sig * tymbalPlate_amplification
              ).clip2 (c_TYMBALPLATE_CLIPPING_AMPLITUDE);


        // model the dominant resonant frequency of the abdominal resonator
        //
        // variables used as constants - based on results from literature for C. australasiae:
        var c_C_AUSTRAL_SPEED_OF_SOUND_M_PER_S
            = 340.0;
        var c_C_AUSTRAL_UNEXTENDED_ABDOMINAL_CAVITY_AVG_VOLUME_M3
            = 1.93 * 1e-06; // verified: yields the 4.56 kHz modeled resonation frequency
        var c_C_AUSTRAL_EXTENDED_ABDOMINAL_CAVITY_EST_AVG_VOLUME_M3
            = 1.09163622611375 * // verified: yields the 4370 Hz observed resonation frequency
              c_C_AUSTRAL_UNEXTENDED_ABDOMINAL_CAVITY_AVG_VOLUME_M3;
        var c_C_AUSTRAL_SINGLE_TYMPANUM_MAX_UNCOVERED_AREA_M2
            = 43.3 * 1e-06;
        var c_C_AUSTRAL_TYMPANA_MAX_UNCOVERED_AREA_M2
            = 2.0 * c_C_AUSTRAL_SINGLE_TYMPANUM_MAX_UNCOVERED_AREA_M2;
        //
        // compute the variably controlled abdominal volume
        //
        // physical modeling here is speculative, based on:
        //
        // - results from literature on C. australasiae:
        //   - abdominal resonation frequency is modeled as a Helmholtz resonator,
        //     i.e. as also based on the volume of the abdominal cavity
        //   - this volume is described as being enlarged as the cicada extends its
        //     abdominal cavity during song production
        //   - values for extended and unextended abdominal volume are either provided
        //     or can be derived
        //
        // - reference video material of H. maculaticollis singing:
        //   - from viewing it seemed possible that, overall, the abdominal cavity
        //     might be less extended during the final phase of the cicada call
        //
        // => the modeled idea:
        //    - have abdominal volume alternate between an extended level during the
        //      initial & middle phase(s), and an unextended level during the final phase
        //    - comparing algorithm output with and without the resulting abdominal volume
        //      drop, its effects seemed hard to discern, both via auditory perception
        //      and in spectrograms
        //
        var mae_mi_on_off_gate
            = SetResetFF.ar (trig: a_mae_trig, reset: a_mi_trig);
        var abdominalVolume_nrmEnv = Env.new
        ( // target levels: from "unextended" (0.0) to "fully extended" (1.0)
          [ 0.0,    1.0,    0.0 ],
          // durations:     for quick transitions between 2 otherwise stable levels
          [       0.005,  0.005 ],
          // curvatures:    make the transitions linear (as an unexamined default)
          [        \lin,   \lin ],
          releaseNode: 1 // "sustain at 2nd level specified, until gate is closed"
        );
        var abdominalVolume_nrmEnv_sig = EnvGen.ar
        ( envelope: abdominalVolume_nrmEnv,
          gate:     mae_mi_on_off_gate
        );
        //
        // normalized abdominal volume, i.e. valid from "unextended" (0.0) to "fully extended" (1.0)
        var abdominal_volume_nrm
            = abdominalVolume_nrmEnv_sig;
        var c_austral_abdominal_volume_m3
            = abdominal_volume_nrm.linlin
              ( 0.0,
                1.0,
                c_C_AUSTRAL_UNEXTENDED_ABDOMINAL_CAVITY_AVG_VOLUME_M3,
                c_C_AUSTRAL_EXTENDED_ABDOMINAL_CAVITY_EST_AVG_VOLUME_M3
              );
        //
        // compute the variably controlled effective tympanal area:
        //
        // physical modeling here is speculative, and based on 2 sources of apparent observations:
        //
        // - from video recordings of singing H. maculaticollis individuals, it seemed that the
        //   start of each phase of a call coincided with movement raising the abdomen, which
        //   afterward was then lowered again;
        // - presumably, such movement, by varying the amount of cover, may respectively increase
        //   then decrease again the effective tympanal area:
        //   - during the initial & (repeated) middle phases of a call: apparently between some
        //     overall maximum and minimum value
        //
        // - from spectrograms of audio recordings, it seemed that during the initial & (repeated)
        //   middle phases of a call, this modulation of effective tympanal area was synchronized
        //   with specific patterns in tymbal muscle contraction frequency
        //
        var c_MAETRIG_TYMPAREA_UNCOVERED_DUR_S = c_MAETRIG_TYMBALMUSCLEFREQ_STARTSEGMENT_DUR_S
                                               + c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S
                                               + c_MAETRIG_TYMBALMUSCLEFREQ_LINRISE_DUR_S
                                               + (c_MAETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5);
        var c_MAETRIG_TYMPAREA_RISE_DUR_S      = c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5;
            // musical choice: made identical to c_MAERETRIG_TYMPAREA_RISE_DUR_S
        var c_MAETRIG_TYMPAREA_PLATEAU_DUR_S   = c_MAETRIG_TYMPAREA_UNCOVERED_DUR_S
                                               - c_MAETRIG_TYMPAREA_RISE_DUR_S;
        var c_MAETRIG_TYMPAREA_FALL_DUR_S      = 0.125 * c_MAEMI_ENVS_DURSCALE;
            // set so as to mimick amplitude profile of an H. maculaticollis reference recording
        var c_MAETRIG_TYMPAREA_2ND_RISE_DUR_S  = 0.335 * c_MAEMI_ENVS_DURSCALE;
            // set so as to mimick amplitude profile of an H. maculaticollis reference recording
        var c_MAETRIG_TYMPAREA_KEEPFLAT_DUR_S  = 999.0;
        var c_MAETRIG_TYMPAREA_MAX_NRM   = 1.0000;
        var c_MAETRIG_TYMPAREA_MIN_NRM   = 0.5997;
            // was tuned to make output harmonically similar to an H. maculaticollis reference recording:
            // - in it, during various sections of insect song, the specific frequency of the currently
            //   strongest harmonic peak could be estimated using spectrum plots
            // - this was done twice for 5 examples of a middle phase within the recorded calls:
            //   once within the segment where effective tympanal area was presumably being maximized
            //   by the cicada, and once where it subsequently was presumably being minimized
            // - the average (fractional) drop in frequency found was then 0.880
            // - the tympanal area minimum was then set so that it made the same drop occur in the value
            //   of the center frequency of abdominal resonation, via abdominal_resonation_freqScaling_nrm
        var c_MAETRIG_TYMPAREA_RANGE_NRM = c_MAETRIG_TYMPAREA_MAX_NRM - c_MAETRIG_TYMPAREA_MIN_NRM;
        var c_MAETRIG_TYMPAREA_START_NRM = c_MAETRIG_TYMPAREA_MIN_NRM
                                           + (c_MAETRIG_TYMPAREA_RANGE_NRM * 0.15);
            // motivation: see c_MAERETRIG_TYMPAREA_START_NRM (identical value)
        //
        var tympArea_maeTrig_mulEnv = Env.new
        ( [ 1.00,
            // target levels
            c_MAETRIG_TYMPAREA_START_NRM,
            c_MAETRIG_TYMPAREA_MAX_NRM,
            c_MAETRIG_TYMPAREA_MAX_NRM,
            c_MAETRIG_TYMPAREA_MIN_NRM,
            c_MAETRIG_TYMPAREA_START_NRM,
            c_MAETRIG_TYMPAREA_START_NRM,
            1.00
          ],
          [ // durations
            0.0,
            c_MAETRIG_TYMPAREA_RISE_DUR_S,
            c_MAETRIG_TYMPAREA_PLATEAU_DUR_S,
            c_MAETRIG_TYMPAREA_FALL_DUR_S,
            c_MAETRIG_TYMPAREA_2ND_RISE_DUR_S,
            c_MAETRIG_TYMPAREA_KEEPFLAT_DUR_S,
            0.0
          ],
          [ // curvatures
            0.0,
            \linear,
            \linear,
            \sine,   // set so as to mimick amplitude profile of an H. maculaticollis reference recording
            \linear,
            \linear,
            0.0
          ]
        );
        var tympArea_maeTrig_mulEnv_sig = EnvGen.ar
        ( envelope: tympArea_maeTrig_mulEnv,
          gate:     a_mae_trig
                    - a_mae_retrig // "on this trigger...
                    - a_mi_trig    //  ...go to final envelope level, immediately"
        );
        //
        var c_MAERETRIG_TYMPAREA_UNCOVERED_DUR_S = c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S
                                                 + c_MAERETRIG_TYMBALMUSCLEFREQ_LINRISE_DUR_S
                                                 + (c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5);
        var c_MAERETRIG_TYMPAREA_RISE_DUR_S      = c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5;
            // estimate based on amplitude profile of an H. maculaticollis reference recording
        var c_MAERETRIG_TYMPAREA_PLATEAU_DUR_S   = c_MAERETRIG_TYMPAREA_UNCOVERED_DUR_S
                                                 - c_MAERETRIG_TYMPAREA_RISE_DUR_S;
        var c_MAERETRIG_TYMPAREA_FALL_DUR_S      = 0.160 * c_MAEMI_ENVS_DURSCALE;
            // set so as to mimick amplitude profile of an H. maculaticollis reference recording
        var c_MAERETRIG_TYMPAREA_2ND_RISE_DUR_S  = 0.195 * c_MAEMI_ENVS_DURSCALE;
            // set so as to mimick amplitude profile of an H. maculaticollis reference recording
        var c_MAERETRIG_TYMPAREA_KEEPFLAT_DUR_S  = 999.0;
        var c_MAERETRIG_TYMPAREA_MAX_NRM   = c_MAETRIG_TYMPAREA_MAX_NRM;
        var c_MAERETRIG_TYMPAREA_MIN_NRM   = c_MAETRIG_TYMPAREA_MIN_NRM;
        var c_MAERETRIG_TYMPAREA_RANGE_NRM = c_MAERETRIG_TYMPAREA_MAX_NRM - c_MAERETRIG_TYMPAREA_MIN_NRM;
        var c_MAERETRIG_TYMPAREA_START_NRM = c_MAERETRIG_TYMPAREA_MIN_NRM
                                           + (c_MAERETRIG_TYMPAREA_RANGE_NRM * 0.15);
            // was set as a trade-off between physical modeling and musical requirements:
            // - mimick the rise seen in the amplitude profile of an H. maculaticollis reference recording
            // - have the attack of middle phases of the insect call not perceptually seem too gradual
            // - avoid the perception of a second attack if the middle phase is used for a longer-than-natural
            //   duration (i.e. longer note)
        //
        var tympArea_maeRetrig_mulEnv = Env.new
        ( [ 1.00,
            // target levels
            c_MAERETRIG_TYMPAREA_START_NRM,
            c_MAERETRIG_TYMPAREA_MAX_NRM,
            c_MAERETRIG_TYMPAREA_MAX_NRM,
            c_MAERETRIG_TYMPAREA_MIN_NRM,
            c_MAERETRIG_TYMPAREA_START_NRM,
            c_MAERETRIG_TYMPAREA_START_NRM,
            1.00
          ],
          [ // durations
            0.0,
            c_MAERETRIG_TYMPAREA_RISE_DUR_S,
            c_MAERETRIG_TYMPAREA_PLATEAU_DUR_S,
            c_MAERETRIG_TYMPAREA_FALL_DUR_S,
            c_MAERETRIG_TYMPAREA_2ND_RISE_DUR_S,
            c_MAERETRIG_TYMPAREA_KEEPFLAT_DUR_S,
            0.0
          ],
          [ // curvatures
            0.0,
            \linear,
            \linear,
            \sine,   // set so as to mimick amplitude profile of an H. maculaticollis reference recording
            \linear,
            \linear,
            0.0
          ]
        );
        var tympArea_maeRetrig_mulEnv_sig = EnvGen.ar
        ( envelope: tympArea_maeRetrig_mulEnv,
          gate:     a_mae_retrig
                    - a_mae_trig // "on this trigger...
                    - a_mi_trig  //  ...go to final envelope level, immediately"
        );
        //
        var c_MITRIG_TYMPAREA_RISE_DUR_S     = c_MAERETRIG_TYMBALMUSCLEFREQ_PEAKLET_DUR_S * 0.5;
            // estimate based on amplitude profile of an H. maculaticollis reference recording
        var c_MITRIG_TYMPAREA_FALL_DUR_S     = 2.361 * c_MAEMI_ENVS_DURSCALE;
            // average of 6 rest-of-final-call-phase durations seen in an H. maculaticollis ref. recording
        var c_MITRIG_TYMPAREA_KEEPFLAT_DUR_S = 999.0;
        var c_MITRIG_TYMPAREA_START_NRM = c_MAERETRIG_TYMPAREA_START_NRM;
        var c_MITRIG_TYMPAREA_END_NRM   = (c_MAERETRIG_TYMPAREA_START_NRM + c_MAERETRIG_TYMPAREA_MAX_NRM) / 2.0;
        var c_MITRIG_TYMPAREA_MAX_NRM   = c_MAETRIG_TYMPAREA_MAX_NRM;
        //
        var tympArea_miTrig_mulEnv = Env.new
        ( [ 1.00,
            // target levels
            c_MITRIG_TYMPAREA_START_NRM, // continue where a middle phase ends
            c_MITRIG_TYMPAREA_MAX_NRM,   // physical modeling choice:
                                         // - rise to the level where in an H. maculaticollis ref. recording,
                                         //   the amplitude profile of final call phases kept at a plateau...
            c_MITRIG_TYMPAREA_END_NRM,   // musical choice:
                                         // - ...however, do not maintain a plateau, but gradually lower
                                         //   tympanal area and thereby also overall amplitude over the
                                         //   duration the final call phase, to a level that makes the
                                         //   attack of a subsequent initial phase more in line with the
                                         //   attack of a middle phase
            c_MITRIG_TYMPAREA_END_NRM,
            1.00
          ],
          [ // durations
            0.0,
            c_MITRIG_TYMPAREA_RISE_DUR_S,
            c_MITRIG_TYMPAREA_FALL_DUR_S,
            c_MITRIG_TYMPAREA_KEEPFLAT_DUR_S,
            0.0
          ],
          [ // curvatures
            0.0,
            \linear,
            \linear,
            \linear,
            0.0
          ]
        );
        var tympArea_miTrig_mulEnv_sig = EnvGen.ar
        ( envelope: tympArea_miTrig_mulEnv,
          gate:     a_mi_trig
                    - a_mae_trig   // "on this trigger...
                    - a_mae_retrig //  ...go to final envelope level, immediately"
        );
        //
        var tympArea_nrmEnv_sig
            = tympArea_maeTrig_mulEnv_sig
            * tympArea_maeRetrig_mulEnv_sig
            * tympArea_miTrig_mulEnv_sig;
        //
        // normalized tympanal area, i.e. valid > "fully covered" (0.0) to "fully uncovered" (1.0):
        var tympanal_area_nrm
            = tympArea_nrmEnv_sig;
        var c_austral_tympanal_area_m2
            = tympanal_area_nrm * c_C_AUSTRAL_TYMPANA_MAX_UNCOVERED_AREA_M2;
        //
        // compute the abdominal resonation frequency that results, according to literature,
        // for C. australasiae: i.e. that of a "bottle-shaped" a.k.a. "Helmholtz" resonator
        var c_austral_single_tympanum_equivalent_radius_m
            = sqrt ((tympanal_area_nrm * c_C_AUSTRAL_SINGLE_TYMPANUM_MAX_UNCOVERED_AREA_M2) / pi);
        var c_austral_tympana_effective_neck_length_m
            = 16.0 * c_austral_single_tympanum_equivalent_radius_m / (3.0 * pi);
        var c_austral_abdominal_resonation_freq_hz
            = (c_C_AUSTRAL_SPEED_OF_SOUND_M_PER_S / (2 * pi)) *
              sqrt
              ( c_austral_tympanal_area_m2 /
                (c_austral_tympana_effective_neck_length_m * c_austral_abdominal_volume_m3)
              );
        //
        // recompute the abdominal resonation frequency as a scaling factor that is relative
        // to the resonation frequency observed during C. australasiae singing
        var abdominal_resonation_freqScaling_nrm
            = c_austral_abdominal_resonation_freq_hz / 4370.0;
        //
        // recompute the abdominal resonation frequency as relative to a musical pitch
        //
        // a musical choice:
        // - initially, this was done so as to implement correct physical modeling:
        //   - in the same way as had been done previously in the causal chain for
        //     - tymbal muscle frequency
        //     - tymbal rib buckling durations
        //     - tymbal rib inter-buckling intervals
        //     - tymbal plate resonation frequency
        //   - that is, by deriving the parameter directly from the currently desired
        //     musical pitch frequency, while otherwise following results from literature
        //     on actual insect song generation
        //   - here, this meant that, when abdominal volume is fully extended and tympanal area
        //     fully uncovered, abdominal resonation should be strongest at the same frequency
        //     as tymbal plate resonation (see commented-out code variant below)
        // - however, an alternative choice also seemed defensible, as an attempt to better
        //   satisfy a musical constraint:
        //   - initial development had been done testing with pitch E5
        //   - when also testing with pitch E4, output from abdominal resonation perceptually
        //     seemed to change not only in pitch, but also in timbre, to such an extent as to
        //     suggest origination from another, different sound source
        //   - this seemed problematic, because such an effect might hinder algorithm output
        //     being perceived as coming from a single voice / stream during melodic use (e.g.
        //     in combination with other voices)
        //   - then, deriving the abdominal resonation frequency from a fixed pitch E5 instead
        //     of from the variable, currently desired pitch seemed to perform better in this
        //     regard
        // - possibly, the above might be an instance of a more general issue:
        //   - physical modeling involving the pitch-based simulation of resonating bodies
        //     in some cases might be expected to produce sonic results that are undesirable
        //     for tonal use precisely when it is successful, in yielding the perception of
        //     differing sound sources for differing pitches
        //
        var abdominal_resonation_freq_hz
          //= abdominal_resonation_freqScaling_nrm * (3.0 * freq_hz); // follow the current musical pitch
            = abdominal_resonation_freqScaling_nrm * (3.0 * c_RAILSBACK_OCT5_E_HZ);
        var abdominal_resonation_freqBased_linAmpComp // compensation for attenuated amplitudes
          //= DC.ar (1.0); // none necessary when following the current musical pitch
            = freq_hz.linlin (c_RAILSBACK_OCT4_E_HZ, c_RAILSBACK_OCT5_E_HZ, 6.946, 0.0).dbamp;
              // determined empirically


        // model the response of the abdominal resonator to the tymbal plate's resonation
        //
        // according to literature sources, this can be done using a narrow bandpass filter;
        // i.e. a "6-pole" (i.e. -36 dB/octave) resonant filter, otherwise similar to the one
        // already used for tymbal plate resonation
        var abdominal_BPF_centreFreq_hz
            = abdominal_resonation_freq_hz;
        //
        // width of the BPF passband in Hz: "RQ" * centre frequency
        // - was tuned to make output harmonically similar to an H. maculaticollis reference recording:
        //   - both there and in algorithm output, the harmonic peaks below insect song frequency,
        //     as estimated and visible in spectrum plots, progressively dropped in amplitude
        //   - the third lower harmonic seemed a representative example of this
        //   - its estimated drop in dB relative to the insect song frequency harmonic
        //     was measured 5 times, each time within the segment of a middle phase of the call
        //     where effective tympanal area was presumably being maximized by the cicada
        //   - the average found was then -10.6 dB
        //   - the BPF RQ value was then set so that algorithm output yielded the same drop
        // - was then verified to also apparently yield the intended perceived pitch:
        //   - testing at pitch E5, the value seemed to fall in a middle ground, between:
        //     - values < approx. 0.5 resulting in a perceived E6/B6
        //     - values > approx. 1.5 resulting in a perceived E4
        var c_ABDOMINAL_BPF_BANDWIDTH_AS_RQ
            = 1.22;
        //
        // the -36 dB/octave bandpass filter is then implemented
        // as a BPF -12 dB/octave filter, 3 times in series
        var unattenuated_abdominal_resonation_sig
            = BPF.ar
              ( in:   BPF.ar
                ( in:   BPF.ar
                  ( in:   tymbalPlate_clippingResonation_sig,
                    freq: abdominal_BPF_centreFreq_hz,
                    rq:   c_ABDOMINAL_BPF_BANDWIDTH_AS_RQ
                  ),
                  freq: abdominal_BPF_centreFreq_hz,
                  rq:   c_ABDOMINAL_BPF_BANDWIDTH_AS_RQ
                ),
                freq: abdominal_BPF_centreFreq_hz,
                rq:   c_ABDOMINAL_BPF_BANDWIDTH_AS_RQ
              )
              * abdominal_resonation_freqBased_linAmpComp;


        // model the overall amplitude attenuation due to the opercula (partially) covering the tympana
        // - inspired by the -11 dB reduction in sound pressure level (SPL) observed in literature for
        //   C. australasiae, when its tympana were fully covered
        //
        var c_MAX_OPERCULAR_ATTENUATION_DB = -38.325;
            // was tuned to make output harmonically similar to an H. maculaticollis reference recording:
            // - in it, during various sections of insect song, the specific amplitude of the currently
            //   strongest harmonic peak could be estimated using spectrum plots
            // - this was done twice for 5 examples of a middle phase within the recorded calls:
            //   once within the segment where effective tympanal area was presumably being maximized
            //   by the cicada, and once where it subsequently was presumably being minimized
            // - the average drop in peak amplitude found was then -15 dB
            // - the maximum opercular attenuation was then set so that, for the value of
            //   c_MAETRIG_TYMPAREA_MIN_NRM found before, it made the same drop in harmonic peak
            //   amplitude occur in algorithm output
        var opercular_attenuation
            = tympanal_area_nrm.linlin (0.0, 1.0, c_MAX_OPERCULAR_ATTENUATION_DB, 0.0).dbamp;
        var abdominal_resonation_sig
            = unattenuated_abdominal_resonation_sig * opercular_attenuation;


        // master_AM_env_sig: amplitude envelope for attenuating overall output & freeing synth
        var master_AM_env = Env.new
        ( [ 1.0,  1.0,      0.0 ], // levels
          [         0,    0.005 ], // durations
          [         0,  \linear ], // curvatures
          releaseNode: 1 // "sustain at 2nd level specified, until gate is closed"
        );
        // add one single-sample impulse to the main gate input signal, as an ugly hack
        // to catch the edge case where it is 0.0 from the start: this would never
        // trigger the envelope, thereby never free the synth on its completion
        var to_master_AM_env_gateSig = a_gate_bit + Impulse.ar (freq: 0);
        var master_AM_env_sig = EnvGen.ar
        ( envelope:   master_AM_env,
          gate:       to_master_AM_env_gateSig,
          doneAction: Done.freeSelf
        );


        // final output
        var audioMonoOut_sig = master_AM_env_sig * abdominal_resonation_sig;
        //
        audioStereoOutAndRec_func.value (audioMonoOut_sig, audioMonoOut_sig);
      }
    ).send (audioSynthesis_group.server);


  } // (end of "soundGeneratingProcesses_init()")


  new_synth
  /* returns a new, running Synth instance of the specified type & initial parametrization
  */
  {
    arg synthType      // a SynthDef name implemented in soundGeneratingProcesses_init()
        ... synthArgs; // zero or more parameters for the new synth

    ^ Synth.new (synthType, synthArgs, target: audioSynthesis_group, addAction: \addToTail);
  }


} // (end of class SoundGeneratingProcesses)



/* This is SuperCollider code, written by Staas de Jong.

   Example of instantiation and real-time parametrization of the \maemi algorithm.

   NB: To run the algorithm, its file ("SoundGeneratingProcesses.sc") should be placed in
       the Extensions directory of your SuperCollider installation, inside two subdirectories,
       as in:

       "YOUR_EXTENSIONS_DIRECTORY\SoundGeneratingProcesses\classes\SoundGeneratingProcesses.sc"
*/


// initialize DSP in general
Server.local.options.numWireBufs    = 512; // more internally complex SynthDefs...
Server.internal.options.numWireBufs = 512; // ...require more "interconnect buffers"
Server.default = ~dsp_server = Server.local.boot;
~rec_buses = Bus.alloc (\audio, ~dsp_server, 2);


// initialize the algorithm (incl. its output & recording buses)
(
  ~sgproc = SoundGeneratingProcesses.new
  ( audioSynthesis_group: ~dsp_server.defaultGroup,
    stereoOut_buses: [0, 1],
    stereoOutAtt_db: -8,
    stereoRec_buses: [~rec_buses.index, ~rec_buses.index + 1],
    stereoRecAtt_db: 0
  );
)


// instantiate the algorithm
~test_synth = ~sgproc.new_synth (\maemi);
~dsp_server.queryAllNodes; // verification: single instance of algorithm now runs on the DSP server?


// real-time control: set the intended musical pitch
(
  ~c_RAILSBACK_OCT5_E_HZ       = 659.73703832729;
  ~c_RAILSBACK_OCT4_E_HZ       = 329.59373351773;
  ~test_synth.set (\freq_hz, ~c_RAILSBACK_OCT5_E_HZ);
)


// real-time control: trig phases of the cicada call
(
  {
    ~test_synth.set (\t_mae_trig, 1.0, \loudness_nrm, 0.4);
    (1.191).wait;
    ~test_synth.set (\t_mae_retrig, 1.0);
    (0.627).wait;
    ~test_synth.set (\t_mae_retrig, 1.0, \loudness_nrm, 0.5);
    (0.619).wait;
    ~test_synth.set (\t_mae_retrig, 1.0, \loudness_nrm, 0.4);
    (0.611).wait;
    ~test_synth.set (\t_mi_trig, 1.0);
  }.fork;
)


// record current algorithm output from internal audio buses
~dsp_server.prepareForRecord (numChannels: ~rec_buses.numChannels);
~dsp_server.record (bus: ~rec_buses, numChannels: ~rec_buses.numChannels);
~dsp_server.stopRecording;


// terminate the algorithm
~test_synth.set (\a_gate_bit, 0);
~test_synth = nil;


// terminate overall DSP
~dsp_server.quit;


https://www.gbif.org/occurrence/2563524032
https://www.gbif.org/occurrence/2563524032
https://species.wikimedia.org/wiki/File:Singing_cicada_audio.ogg
https://species.wikimedia.org/wiki/File:Singing_cicada_audio.ogg
https://commons.wikimedia.org/wiki/File:Hyalessa_maculaticollis_v01.ogg
https://commons.wikimedia.org/wiki/File:Hyalessa_maculaticollis_v01.ogg

